
  `PM: CNFG_DealQuickEntry_Init

  `The Object DLOG_PrefsQEHandle was created in .gpPreferences
ALP_ALPAreaSet (DLOG_PrefsQEHandle;eDealQuickEntry)

  `Init the arrays that are used in the Editor
CNFG_CnfgFormArrays_Init (0)

  `Create default arrays
$DefaultArrayHandle:=OT_ObjectCreate 
CNFG_CnfgFormDfltArrays_Crt ($DefaultArrayHandle;DLOG_PrefsQEHandle)

  `Get the Admin record
CNFG_ConfigTypeSet (DLOG_PrefsQEHandle;"DealQuickEntry")
CNFG_CnfgFormRecGet (DLOG_PrefsQEHandle)

  `Get the Object from the BLOB
$ConfigHandle:=OT_BLOBToObject (->[User]CONFIG_BLOB)

If (Not(OT_ObjectHandleIsValid ($ConfigHandle)))

$ArrayHandle:=OT_ObjectCreate 
$SettingsHandle:=OT_ObjectCreate 

Else 
  `Get the arrays that have been stored in the Object
$ArrayHandle:=OT_ObjectGet ($ConfigHandle;"$ArrayHandle")

  `Get the settings that have been stored in the Object
$SettingsHandle:=OT_ObjectGet ($ConfigHandle;"$SettingsHandle")
OT_ObjectClear ($ConfigHandle)

End if   `If (Not(OT_ObjectHandleIsValid ))

$MandatoryArrayHandle:=OT_ObjectCreate 
CNFG_CnfgFormMandatoryFieldsSet ($MandatoryArrayHandle)

  `Fill the display arrays
CNFG_CnfgFormArrays_Fill ($ArrayHandle;$DefaultArrayHandle;$MandatoryArrayHandle)
OT_ObjectClear ($ArrayHandle)
OT_ObjectClear ($DefaultArrayHandle)

CNFG_QEShowALPCheckboxValsGet ($SettingsHandle)

◊CNFG_OpenFromNewMenuItem_L:=CNFG_QEFromNewGet ($SettingsHandle)

OT_ObjectPut (DLOG_PrefsQEHandle;"$SettingsHandle";$SettingsHandle)

OT_ObjectPut (DLOG_PrefsQEHandle;"$MandatoryArrayHandle";$MandatoryArrayHandle)

CNFG_QEShowALPCheckboxValsSet (DLOG_PrefsQEHandle)
$MaxFields:=CNFG_MaxQEFieldsCalc (DLOG_PrefsQEHandle)
CNFG_MaxQEFieldsSet (DLOG_PrefsQEHandle;$MaxFields)

OT_pCheckMarkArraySet (DLOG_PrefsQEHandle;->CNFG_CheckMark_AT)
OT_pFieldNameArraySet (DLOG_PrefsQEHandle;->CNFG_FieldName_AT)
OT_pSequenceNumberArraySet (DLOG_PrefsQEHandle;->CNFG_SequenceNumber_AL)
OT_pFieldIDArraySet (DLOG_PrefsQEHandle;->CNFG_FieldID_AT)

◊CNFG_FieldNumberLimit_T:=CNFG_FieldLimitStringCreate (DLOG_PrefsQEHandle)

$NumFieldsSelected:=UTIL_ARR_ValuesCount_Text (->CNFG_CheckMark_AT;◊ALCheckmark)
CNFG_NumFieldsSelectedSet (DLOG_PrefsQEHandle;$NumFieldsSelected)

  `Load the arrays into the AL area
If (CNFG_CnfgFormALP_ArraysLoad (DLOG_PrefsQEHandle)=0)

CNFG_CnfgFormALP_Format (DLOG_PrefsQEHandle)
Else 

gpAlert ("Invalid ALP call in CNFG_DealQuickEntry_Init.")
End if 

OT_ObjectClear ($SettingsHandle)
OT_ObjectClear ($MandatoryArrayHandle)

  `eom


